
1.1 Change 1

Testcase name
7.2.3.9

Reason for change
1. In TDD while sending the periodic UL grant the period should be the multiple of 5 as per the latest test specification.

2. The current TTCN configures SS to transmit grant every UL TTI (1 ms). This can result in test case failure due to HARQ retransmission errors during the test case execution when no data in expected in the UL. The default UL grant allocation can be used to assign grant as when UE has any UL data to transmit. After step 3, 8A and 10A the SS should disable the continuous UL grant transmission.

Summary of change
1. Send the periodic UL grant with period set to 5.

2. Start default UL grant after step 3, 8A and 10A.



Source of change
RLC_AM_Testcases.ttcn

Before:

function fl_EUTRA_7_2_3_9_TestBody(inout RLC_SS_State p_RLC_Rec) runs on EUTRA_PTC

{ // @sic R5-105016 sic@

  var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

  var template (value) EUTRA_FDD_TDD_CellInfo_Type  v_EUTRA_FDD_TDD_Info;

  var integer i;

  var DRB_COMMON_IND v_ReceivedAsp;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  var template (value) DRB_DataPerSubframeList_DL_Type v_SubframeDataList;

  var B1_Type v_Poll;

  var integer v_Duration;

  var SubFrameTiming_Type v_SubFrameTimingTa;

  var SubFrameTiming_Type v_SubFrameTimingTb;

  var template (value) RLC_Status_NACK_List_Type   v_NACK_List;

  var RLC_Status_Padding_Type  v_Padding_5bits := '00000'B;

  var SubFrameTiming_Type v_Timing;

  var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode ;

  var template (value) PhysicalConfigDedicated_AntennaInfo_Type   v_AntennaInfo;

  timer t_WatchDog := 5.0;

  // determine RAT to make the timing correct later on

  v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

  // get FDD_TDD_Info

  v_EUTRA_FDD_TDD_Info:=f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_Cell1);

  v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_Cell1);

  //PDCP SDUs of this size will be built, appended to a PDCP header,

  //and then used as RLC SDUs

  p_RLC_Rec.TxDataSize := 318; // 318 octets = (2560 bits - 16 bits) / 8

  // generate 172 RLC SDUs = PDCP PDUs for transmission

  for (i := 0; i < 172; i:= i + 1) {

    f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

    v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN(int2bit(i, tsc_PDCP_SN_Size_12 ), p_RLC_Rec.RLC_Data);

    p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod (tsc_Maximum_PDCP_SN_12 + 1);

  }

  //@siclog "Step 0" siclog@

  // The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);

  //@siclog "Step 1" siclog@

  // The SS transmits 4 AMD PDUs, each containing an RLC SDU of size 2560 bits.

  // generate 4 RLC PDUs

  for (i := 0; i < 4; i:= i + 1) {

    if ((i+1) == tsc_AM_Window_Size) {

      v_Poll := tsc_P_Poll;

    }

    else {

      v_Poll := tsc_P_NoPoll;

    }

    v_SubframeDataList[i].SubframeOffset := f_EUTRA_DataListSubframeOffset(v_EUTRA_FDD_TDD_Mode, i);

    v_SubframeDataList[i].HarqProcess := {Automatic := true};

    v_SubframeDataList[i].PduSduList := { RlcPdu := {{AMD := cs_AMD_PDU_NoLIs (p_RLC_Rec.AM_VTS,

                                                                               v_Poll,

                                                                               tsc_FI_FullSDU,

                                                                               {p_RLC_Rec.RLC_DataList[i]} )}}};

    p_RLC_Rec.AM_VTS := (p_RLC_Rec.AM_VTS + 1) mod 1024;

  }

  // send the list of 4 RLC PDUs

  // at least 100 ms to wait for having an accurate reference to start

  v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

  DRB.send (cas_DRB1_RLCAmDataReq_List (eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                        v_SubframeDataList));

  //@siclog "Step 1A" siclog@

  // The SS waits for 60 ms, then starts assigning UL grants (UL grant allocation type 2) of size 2600 bits. (Note 1)

  v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 60);

  f_EUTRA_PeriodicULGrantTransmission( eutra_Cell1,

                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                       15, 9, 1); // Imcs = 15 -> Itbs = 14
  // EXCEPTION: In parallel to the events described in step 1, the steps specified in Table 7.2.3.9.3.2-2 should take place

  // Check: Does the UE transmit 4 AMD PDUs, with only the last one having the poll bit set? Record time TA when the PDU with the poll bit set is received at SS.

  // Record time TA when the PDU with the poll bit set is received at SS.

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU1);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU2);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU3);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU4);

  v_SubFrameTimingTa := p_RLC_Rec.TimeStampLastReceivedPDU;

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Parallel behaviour 7.2.3.9.3.2-2");

  //@siclog "Step 2" siclog@

  // Check1: does the UE transmit an AMD PDU with with a SN in range 0 to 3 and and P=1?

  // Record time TB.

  // Check 2: (TB - TA) = t-PollRetransmit.

  DRB.receive (car_DRB1_RLCAmDataInd (eutra_Cell1,

                                      cr_AMD_PDUoneSDU_Any((int2bit (0, tsc_AM_SN_Size),

                                                            int2bit (1, tsc_AM_SN_Size),

                                                            int2bit (2, tsc_AM_SN_Size),

                                                            int2bit (3, tsc_AM_SN_Size)),

                                                           tsc_P_Poll,

                                                           ? ))) -> value v_ReceivedAsp;

  v_SubFrameTimingTb := v_ReceivedAsp.Common.TimingInfo.SubFrame;

  v_Duration := f_EUTRA_SubFrameTimingDuration( v_SubFrameTimingTa, v_SubFrameTimingTb);

  if ( f_EUTRA_SubFrameTimingCheckDuration( eutra_Cell1, l2Timer, v_Duration, float2int(p_RLC_Rec.t_PollRetransmit * 1000.0) ) )

    {

      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    }

  else

    {

      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2");

    }

  //@siclog "Step 3" siclog@

  // Upon receiving the Poll, the SS transmits an RLC Status Report.

  f_TxSTATUS_PDU(p_RLC_Rec);

  //@siclog "Step 4" siclog@

  // Check: does the UE retransmit an AMD PDU within 1 sec ?

  t_WatchDog.start (1.0);

  alt {

    [] t_WatchDog.timeout

      {

      }

    [] DRB.receive (car_DRB1_RLCAmDataInd(eutra_Cell1, ?)) // catches any SDU, not only SDU4

      {

        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");

      }

  } // alt

  //@siclog "Step 5" siclog@

  // SS performs an RRC Connection Reconfiguration procedure changing pollPDU to p4.

  f_EUTRA_RRCConnectionReconfiguration_RadioResConfig(eutra_Cell1,

                                                      v_RRC_TI,

                                                      cs_SRB2_1AM_DRB_Reconfig_RLC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                   v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                   v_AntennaInfo,

                                                                                   cds_TX_AM_RLC_no_mRT (ms400, p4, kB25),

                                                                                   cs_508_DRB_DL_AM_RLC));

  p_RLC_Rec.pollPDU := 4; // p4

  //@siclog "Step 5A" siclog@

  // The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);

  //@siclog "Step 5B" siclog@

  // The SS allows one HARQ retransmission from now on.

  f_EUTRA_HarqErrorHandling_Set(allowOneRetransmission);

  //@siclog "Step 6" siclog@

  // The SS transmits 8 AMD PDUs such that 1 AMD PDU is sent every second radio frame,

  // each containing an RLC SDU of 2 560 bits. (Note 2)

  // generate 8 RLC PDUs, next 8 of the SDUs generated earlier

  for (i := 0; i < 8; i:= i + 1) {

    if ((i+1) == tsc_AM_Window_Size) {

      v_Poll := tsc_P_Poll;

    }

    else {

      v_Poll := tsc_P_NoPoll;

    }

    v_SubframeDataList[i].SubframeOffset := (20 * (i mod 512));

    v_SubframeDataList[i].HarqProcess := {Automatic := true};

    v_SubframeDataList[i].PduSduList := { RlcPdu := {{AMD := cs_AMD_PDU_NoLIs (p_RLC_Rec.AM_VTS,

                                                                               v_Poll,

                                                                               tsc_FI_FullSDU,

                                                                               {p_RLC_Rec.RLC_DataList[i+4]} )}}};

    p_RLC_Rec.AM_VTS := (p_RLC_Rec.AM_VTS + 1) mod 1024;

  }

  // send the list of 8 RLC PDUs

  // at least 100 ms to wait for having an accurate reference to start

  v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

  DRB.send (cas_DRB1_RLCAmDataReq_List( eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                        v_SubframeDataList));

  //@siclog "Step 6A" siclog@

  // The SS waits for 60 ms after the first DL AMD PDU has been transmitted in step 6, then starts assigning

  // UL grants (UL grant allocation type 2) in every second radio frame of size 2600 bits. (Note 1)(Note 2)

  v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 60);

  f_EUTRA_PeriodicULGrantTransmission( eutra_Cell1,

                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                       15, 9, 20); // Imcs = 15 -> Itbs = 14

  // EXCEPTION: In parallel to the events described in step 6, the steps specified in Table 7.2.3.9.3.2-3

  // should take place.

  // Check: Does the UE transmit 8 AMD PDUs, with the poll bit set only in the 4th and the 8th PDUs?

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU5);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU6);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU7);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_Poll,   tsc_RLC_SDU8);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU9);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU10);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU11);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_Poll,   tsc_RLC_SDU12);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Parallel behaviour 7.2.3.9.3.2-3");

  //@siclog "Step 7" siclog@

  // The SS sends a Status Report with NACK_SN=0, NACK_SN=1, NACK_SN=2, NACK_SN=4 and NACK_SN=5.

  v_NACK_List := cs_NACK_List_5NACKs_NoSOs(4, 5, 6, 8, 9);

  f_TxSTATUS_PDU_NACK( p_RLC_Rec, v_NACK_List, v_Padding_5bits); // @sic R5s100531 sic@

  //@siclog "Step 8" siclog@

  // Check: does the UE transmit AMD PDUs with the following SN and P values:

  // AMD PDU, SN=4, P=0

  // AMD PDU, SN=5, P=0

  // AMD PDU, SN=6, P=0

  // AMD PDU, SN=8, P=0

  // AMD PDU, SN=9, P=1

  p_RLC_Rec.AM_VRR := 4;

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU5);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU6);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU7);

  p_RLC_Rec.AM_VRR := p_RLC_Rec.AM_VRR + 1;

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU9);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU10);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");

  //@siclog "Step 8A" siclog@

  // The SS transmits a Status Report with ACK_SN=12 and no NACK_SN.

  p_RLC_Rec.AM_VRR := 12;

  f_TxSTATUS_PDU(p_RLC_Rec);

  //@siclog "Step 9" siclog@

  // SS performs a RRC Connection Reconfiguration procedure triggering an RLC-reestablishment

  // of the RLC entity and sets pollPDU to 256.

  // reconfigure DRB at the UE side

  f_EUTRA_RRCConnectionReconfiguration_RadioResConfig(eutra_Cell1,

                                                      v_RRC_TI,

                                                      cs_SRB2_1AM_DRB_Reconfig_RLC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                   v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                   v_AntennaInfo,

                                                                                   cds_TX_AM_RLC_no_mRT (ms400, p256, kB25),

                                                                                   cs_508_DRB_DL_AM_RLC));

  p_RLC_Rec.pollPDU := 256; // p256

  //@siclog "Step 9A" siclog@

  // The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);

  //@siclog "Step 10" siclog@

  // After 500 ms the SS transmits 160 AMD PDUs such that 1 AMD PDU is sent every second radio frame,

  // each containing an RLC SDU of size 2560 bits. (Note 2)

  // generate 160 RLC PDUs, next 160 of the SDUs generated earlier

  for (i := 0; i < 160; i:= i + 1) {

    if ((i+1) == tsc_AM_Window_Size) {

      v_Poll := tsc_P_Poll;

    }

    else {

      v_Poll := tsc_P_NoPoll;

    }

    v_SubframeDataList[i].SubframeOffset := (20 * (i mod 512));

    v_SubframeDataList[i].HarqProcess := {Automatic := true};

    v_SubframeDataList[i].PduSduList := { RlcPdu := {{AMD := cs_AMD_PDU_NoLIs (p_RLC_Rec.AM_VTS,

                                                                               v_Poll,

                                                                               tsc_FI_FullSDU,

                                                                               {p_RLC_Rec.RLC_DataList[i+12]} )}}};

    p_RLC_Rec.AM_VTS := (p_RLC_Rec.AM_VTS + 1) mod 1024;

  }

  // send the list of 160 RLC PDUs

  v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1, 500);

  DRB.send (cas_DRB1_RLCAmDataReq_List( eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                        v_SubframeDataList));

  //@siclog "Step 10A" siclog@

  // The SS waits for 60 ms after the first DL AMD PDU has been transmitted in step 10, then starts assigning

  // UL grants (UL grant allocation type 2) in every second radio frame of size 2600 bits. (Note 1)(Note 2)

  v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 60);

  f_EUTRA_PeriodicULGrantTransmission( eutra_Cell1,

                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                       15, 9, 20); // Imcs = 15 -> Itbs = 14

  // EXCEPTION: In parallel to the events described in step 10, the steps specified in Table 7.2.3.9.3.2-4 should take place

  // Check: Does the UE transmit 160 AMD PDUs, with the poll bit set only in the 79th one, the 158th one and the 160th one?

  for (i := 0; i < 160; i:= i + 1) {

    if (((i+1) == 79) or ((i+1) == 158) or ((i+1) == 160)){

      f_RxAMD_SDU (p_RLC_Rec, tsc_P_Poll, i+12);

      // The SS transmits an RLC Status Report.

      f_TxSTATUS_PDU(p_RLC_Rec);

      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Parallel behaviour 7.2.3.9.3.2-4");

    }

    else {

      f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, i+12);

    }

  }

} // fl_EUTRA_7_2_3_9_TestBody

After:

function fl_EUTRA_7_2_3_9_TestBody(inout RLC_SS_State p_RLC_Rec) runs on EUTRA_PTC

{ // @sic R5-105016 sic@

  var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

  var template (value) EUTRA_FDD_TDD_CellInfo_Type  v_EUTRA_FDD_TDD_Info;

  var integer i;

  var DRB_COMMON_IND v_ReceivedAsp;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  var template (value) DRB_DataPerSubframeList_DL_Type v_SubframeDataList;

  var B1_Type v_Poll;

  var integer v_Duration;

  var SubFrameTiming_Type v_SubFrameTimingTa;

  var SubFrameTiming_Type v_SubFrameTimingTb;

  var template (value) RLC_Status_NACK_List_Type   v_NACK_List;

  var RLC_Status_Padding_Type  v_Padding_5bits := '00000'B;

  var SubFrameTiming_Type v_Timing;

  var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode ;

  var template (value) PhysicalConfigDedicated_AntennaInfo_Type   v_AntennaInfo;

  timer t_WatchDog := 5.0;

  // determine RAT to make the timing correct later on

  v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

  // get FDD_TDD_Info

  v_EUTRA_FDD_TDD_Info:=f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_Cell1);

  v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_Cell1);

  //PDCP SDUs of this size will be built, appended to a PDCP header,

  //and then used as RLC SDUs

  p_RLC_Rec.TxDataSize := 318; // 318 octets = (2560 bits - 16 bits) / 8

  // generate 172 RLC SDUs = PDCP PDUs for transmission

  for (i := 0; i < 172; i:= i + 1) {

    f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

    v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN(int2bit(i, tsc_PDCP_SN_Size_12 ), p_RLC_Rec.RLC_Data);

    p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod (tsc_Maximum_PDCP_SN_12 + 1);

  }

  //@siclog "Step 0" siclog@

  // The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);

  //@siclog "Step 1" siclog@

  // The SS transmits 4 AMD PDUs, each containing an RLC SDU of size 2560 bits.

  // generate 4 RLC PDUs

  for (i := 0; i < 4; i:= i + 1) {

    if ((i+1) == tsc_AM_Window_Size) {

      v_Poll := tsc_P_Poll;

    }

    else {

      v_Poll := tsc_P_NoPoll;

    }

    v_SubframeDataList[i].SubframeOffset := f_EUTRA_DataListSubframeOffset(v_EUTRA_FDD_TDD_Mode, i);

    v_SubframeDataList[i].HarqProcess := {Automatic := true};

    v_SubframeDataList[i].PduSduList := { RlcPdu := {{AMD := cs_AMD_PDU_NoLIs (p_RLC_Rec.AM_VTS,

                                                                               v_Poll,

                                                                               tsc_FI_FullSDU,

                                                                               {p_RLC_Rec.RLC_DataList[i]} )}}};

    p_RLC_Rec.AM_VTS := (p_RLC_Rec.AM_VTS + 1) mod 1024;

  }

  // send the list of 4 RLC PDUs

  // at least 100 ms to wait for having an accurate reference to start

  v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

  DRB.send (cas_DRB1_RLCAmDataReq_List (eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                        v_SubframeDataList));

  //@siclog "Step 1A" siclog@

  // The SS waits for 60 ms, then starts assigning UL grants (UL grant allocation type 2) of size 2600 bits. (Note 1)

  v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 60);

  if (v_EUTRA_FDD_TDD_Mode == FDD)

  {  

     f_EUTRA_PeriodicULGrantTransmission( eutra_Cell1,

                                          cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                          15, 9, 1); // Imcs = 15 -> Itbs = 14

  }

  else

  {

    f_EUTRA_PeriodicULGrantTransmission( eutra_Cell1,

                                         cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                         15, 9, 5); // Imcs = 15 -> Itbs = 14

  }   // NOTE 1                                         

  // EXCEPTION: In parallel to the events described in step 1, the steps specified in Table 7.2.3.9.3.2-2 should take place

  // Check: Does the UE transmit 4 AMD PDUs, with only the last one having the poll bit set? Record time TA when the PDU with the poll bit set is received at SS.

  // Record time TA when the PDU with the poll bit set is received at SS.

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU1);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU2);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU3);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU4);

  v_SubFrameTimingTa := p_RLC_Rec.TimeStampLastReceivedPDU;

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Parallel behaviour 7.2.3.9.3.2-2");

  //@siclog "Step 2" siclog@

  // Check1: does the UE transmit an AMD PDU with with a SN in range 0 to 3 and and P=1?

  // Record time TB.

  // Check 2: (TB - TA) = t-PollRetransmit.

  DRB.receive (car_DRB1_RLCAmDataInd (eutra_Cell1,

                                      cr_AMD_PDUoneSDU_Any((int2bit (0, tsc_AM_SN_Size),

                                                            int2bit (1, tsc_AM_SN_Size),

                                                            int2bit (2, tsc_AM_SN_Size),

                                                            int2bit (3, tsc_AM_SN_Size)),

                                                           tsc_P_Poll,

                                                           ? ))) -> value v_ReceivedAsp;

  v_SubFrameTimingTb := v_ReceivedAsp.Common.TimingInfo.SubFrame;

  v_Duration := f_EUTRA_SubFrameTimingDuration( v_SubFrameTimingTa, v_SubFrameTimingTb);

  if ( f_EUTRA_SubFrameTimingCheckDuration( eutra_Cell1, l2Timer, v_Duration, float2int(p_RLC_Rec.t_PollRetransmit * 1000.0) ) )

    {

      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    }

  else

    {

      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2");

    }

  //@siclog "Step 3" siclog@

  // Upon receiving the Poll, the SS transmits an RLC Status Report.

  f_TxSTATUS_PDU(p_RLC_Rec);

  f_EUTRA_StartDefULGrantTransmission(eutra_Cell1, cs_TimingInfo_Now);   // NOTE 2                                        

  //@siclog "Step 4" siclog@

  // Check: does the UE retransmit an AMD PDU within 1 sec ?

  t_WatchDog.start (1.0);

  alt {

    [] t_WatchDog.timeout

      {

      }

    [] DRB.receive (car_DRB1_RLCAmDataInd(eutra_Cell1, ?)) // catches any SDU, not only SDU4

      {

        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");

      }

  } // alt

  //@siclog "Step 5" siclog@

  // SS performs an RRC Connection Reconfiguration procedure changing pollPDU to p4.

  f_EUTRA_RRCConnectionReconfiguration_RadioResConfig(eutra_Cell1,

                                                      v_RRC_TI,

                                                      cs_SRB2_1AM_DRB_Reconfig_RLC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                   v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                   v_AntennaInfo,

                                                                                   cds_TX_AM_RLC_no_mRT (ms400, p4, kB25),

                                                                                   cs_508_DRB_DL_AM_RLC));

  p_RLC_Rec.pollPDU := 4; // p4

  //@siclog "Step 5A" siclog@

  // The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);

  //@siclog "Step 5B" siclog@

  // The SS allows one HARQ retransmission from now on.

  f_EUTRA_HarqErrorHandling_Set(allowOneRetransmission);

  //@siclog "Step 6" siclog@

  // The SS transmits 8 AMD PDUs such that 1 AMD PDU is sent every second radio frame,

  // each containing an RLC SDU of 2 560 bits. (Note 2)

  // generate 8 RLC PDUs, next 8 of the SDUs generated earlier

  for (i := 0; i < 8; i:= i + 1) {

    if ((i+1) == tsc_AM_Window_Size) {

      v_Poll := tsc_P_Poll;

    }

    else {

      v_Poll := tsc_P_NoPoll;

    }

    v_SubframeDataList[i].SubframeOffset := (20 * (i mod 512));

    v_SubframeDataList[i].HarqProcess := {Automatic := true};

    v_SubframeDataList[i].PduSduList := { RlcPdu := {{AMD := cs_AMD_PDU_NoLIs (p_RLC_Rec.AM_VTS,

                                                                               v_Poll,

                                                                               tsc_FI_FullSDU,

                                                                               {p_RLC_Rec.RLC_DataList[i+4]} )}}};

    p_RLC_Rec.AM_VTS := (p_RLC_Rec.AM_VTS + 1) mod 1024;

  }

  // send the list of 8 RLC PDUs

  // at least 100 ms to wait for having an accurate reference to start

  v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

  DRB.send (cas_DRB1_RLCAmDataReq_List( eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                        v_SubframeDataList));

  //@siclog "Step 6A" siclog@

  // The SS waits for 60 ms after the first DL AMD PDU has been transmitted in step 6, then starts assigning

  // UL grants (UL grant allocation type 2) in every second radio frame of size 2600 bits. (Note 1)(Note 2)

  v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 60);

  f_EUTRA_PeriodicULGrantTransmission( eutra_Cell1,

                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                       15, 9, 20); // Imcs = 15 -> Itbs = 14

  // EXCEPTION: In parallel to the events described in step 6, the steps specified in Table 7.2.3.9.3.2-3

  // should take place.

  // Check: Does the UE transmit 8 AMD PDUs, with the poll bit set only in the 4th and the 8th PDUs?

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU5);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU6);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU7);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_Poll,   tsc_RLC_SDU8);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU9);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU10);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU11);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_Poll,   tsc_RLC_SDU12);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Parallel behaviour 7.2.3.9.3.2-3");

  //@siclog "Step 7" siclog@

  // The SS sends a Status Report with NACK_SN=0, NACK_SN=1, NACK_SN=2, NACK_SN=4 and NACK_SN=5.

  v_NACK_List := cs_NACK_List_5NACKs_NoSOs(4, 5, 6, 8, 9);

  f_TxSTATUS_PDU_NACK( p_RLC_Rec, v_NACK_List, v_Padding_5bits); // @sic R5s100531 sic@

  //@siclog "Step 8" siclog@

  // Check: does the UE transmit AMD PDUs with the following SN and P values:

  // AMD PDU, SN=4, P=0

  // AMD PDU, SN=5, P=0

  // AMD PDU, SN=6, P=0

  // AMD PDU, SN=8, P=0

  // AMD PDU, SN=9, P=1

  p_RLC_Rec.AM_VRR := 4;

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU5);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU6);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU7);

  p_RLC_Rec.AM_VRR := p_RLC_Rec.AM_VRR + 1;

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU9);

  f_RxAMD_SDU (p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU10);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");

  //@siclog "Step 8A" siclog@

  // The SS transmits a Status Report with ACK_SN=12 and no NACK_SN.

  p_RLC_Rec.AM_VRR := 12;

  f_TxSTATUS_PDU(p_RLC_Rec);

  f_EUTRA_StartDefULGrantTransmission(eutra_Cell1, cs_TimingInfo_Now);   // NOTE 2                                        

  //@siclog "Step 9" siclog@

  // SS performs a RRC Connection Reconfiguration procedure triggering an RLC-reestablishment

  // of the RLC entity and sets pollPDU to 256.

  // reconfigure DRB at the UE side

  f_EUTRA_RRCConnectionReconfiguration_RadioResConfig(eutra_Cell1,

                                                      v_RRC_TI,

                                                      cs_SRB2_1AM_DRB_Reconfig_RLC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                   v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                   v_AntennaInfo,

                                                                                   cds_TX_AM_RLC_no_mRT (ms400, p256, kB25),

                                                                                   cs_508_DRB_DL_AM_RLC));

  p_RLC_Rec.pollPDU := 256; // p256

  //@siclog "Step 9A" siclog@

  // The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);

  //@siclog "Step 10" siclog@

  // After 500 ms the SS transmits 160 AMD PDUs such that 1 AMD PDU is sent every second radio frame,

  // each containing an RLC SDU of size 2560 bits. (Note 2)

  // generate 160 RLC PDUs, next 160 of the SDUs generated earlier

  for (i := 0; i < 160; i:= i + 1) {

    if ((i+1) == tsc_AM_Window_Size) {

      v_Poll := tsc_P_Poll;

    }

    else {

      v_Poll := tsc_P_NoPoll;

    }

    v_SubframeDataList[i].SubframeOffset := (20 * (i mod 512));

    v_SubframeDataList[i].HarqProcess := {Automatic := true};

    v_SubframeDataList[i].PduSduList := { RlcPdu := {{AMD := cs_AMD_PDU_NoLIs (p_RLC_Rec.AM_VTS,

                                                                               v_Poll,

                                                                               tsc_FI_FullSDU,

                                                                               {p_RLC_Rec.RLC_DataList[i+12]} )}}};

    p_RLC_Rec.AM_VTS := (p_RLC_Rec.AM_VTS + 1) mod 1024;

  }

  // send the list of 160 RLC PDUs

  v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1, 500);

  DRB.send (cas_DRB1_RLCAmDataReq_List( eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                        v_SubframeDataList));

  //@siclog "Step 10A" siclog@

  // The SS waits for 60 ms after the first DL AMD PDU has been transmitted in step 10, then starts assigning

  // UL grants (UL grant allocation type 2) in every second radio frame of size 2600 bits. (Note 1)(Note 2)

  v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 60);

  f_EUTRA_PeriodicULGrantTransmission( eutra_Cell1,

                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                       15, 9, 20); // Imcs = 15 -> Itbs = 14

  // EXCEPTION: In parallel to the events described in step 10, the steps specified in Table 7.2.3.9.3.2-4 should take place

  // Check: Does the UE transmit 160 AMD PDUs, with the poll bit set only in the 79th one, the 158th one and the 160th one?

  for (i := 0; i < 160; i:= i + 1) {

    if (((i+1) == 79) or ((i+1) == 158) or ((i+1) == 160)){

      f_RxAMD_SDU (p_RLC_Rec, tsc_P_Poll, i+12);

      // The SS transmits an RLC Status Report.

      f_TxSTATUS_PDU(p_RLC_Rec);

      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Parallel behaviour 7.2.3.9.3.2-4");

    }

    else {

      f_RxAMD_SDU (p_RLC_Rec, tsc_P_NoPoll, i+12);

    }

  }

  f_EUTRA_StartDefULGrantTransmission(eutra_Cell1, cs_TimingInfo_Now);   // NOTE 2                                        

} // fl_EUTRA_7_2_3_9_TestBody

